// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _setOwner(_msgSender());
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _setOwner(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _setOwner(newOwner);
    }

    function _setOwner(address newOwner) private {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/vesting/interfaces/IVesting.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.10;

/**
 * @title An interface for the Vesting contract.
 * @author Asymetrix Protocol Inc Team
 * @notice An interface that describes data structures and events for the
 *         Vesting contract.
 */
interface IVesting {
    /**
     * @notice Vesting schedule structure.
     * @param amount An amount of ASX tokens to vest.
     * @param released An amount of ASX tokens that already were released.
     * @param owner An owner of a vesting schedule.
     * @param startTimestamp A timestamp when a vesting schedule started.
     * @param lockPeriod A lock period duration (in seconds) that should have
     *                   place before distribution will start.
     * @param releasePeriod A period (in seconds) during wich ASX tokens will
     *                      be distributed after the lock period.
     */
    struct VestingSchedule {
        uint256 amount;
        uint256 released;
        address owner;
        uint32 startTimestamp;
        uint32 lockPeriod;
        uint32 releasePeriod;
    }

    /**
     * @notice Event emitted when vesting schedule was created.
     * @param vestingSchedule A newly created vesting schedule object.
     */
    event VestingScheduleCreated(VestingSchedule vestingSchedule);

    /**
     * @notice Event emitted when a part of ASX tokens from a vesting schedule
     *         was released.
     * @param vsid An ID of a vesting schedule.
     * @param recipient An address of ASX tokens recipient.
     * @param amount An amount of tokens that were released.
     */
    event Released(
        uint256 indexed vsid,
        address indexed recipient,
        uint256 amount
    );

    /**
     * @notice Event emitted when a part of unused ASX tokens or other tokens
     *         (including ETH) was withdrawn by an owner.
     * @param token A token that was withdraw. If token address was equal to
     *              zero address - ETH were withdrawn.
     * @param owner An address of an owner that withdrawn unused tokens.
     * @param amount An amount of tokens that were withdrawn.
     */
    event Withdrawn(
        address indexed token,
        address indexed owner,
        uint256 amount
    );
}

// File: contracts/vesting/Vesting.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity 0.8.10;

import "./interfaces/IVesting.sol";

import "@openzeppelin/contracts/utils/Address.sol";

import "@openzeppelin/contracts/access/Ownable.sol";

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";

/**
 * @title Vesting contract.
 * @author Asymetrix Protocol Inc Team
 * @notice An implementation of a Vesting contract for ASX token distribution
 *         using vesting schedules.
 */
contract Vesting is Ownable, IVesting {
    using SafeERC20 for IERC20;
    using Address for address;

    mapping(uint256 => VestingSchedule) private vestingSchedules;

    IERC20 private immutable token;

    uint256 private vestingSchedulesCount;

    uint256 private totalDistributionAmount;

    uint256 private totalReleasedAmount;

    /**
     * @notice Сonstructor of this Vesting contract.
     * @dev Sets ASX token contract address.
     * @param _token ASX token contract address.
     */
    constructor(address _token) {
        require(_token.isContract(), "Vesting: invalid ASX token address");

        token = IERC20(_token);
    }

    /**
     * @notice Returns the ASX token address.
     * @return The ASX token address.
     */
    function getToken() external view returns (address) {
        return address(token);
    }

    /**
     * @notice Returns total vesting schedules count.
     * @return Total vesting schedules count.
     */
    function getVestingSchedulesCount() external view returns (uint256) {
        return vestingSchedulesCount;
    }

    /**
     * @notice Returns total distribution amount for all vesting schedules.
     * @return Total distribution amount for all vesting schedules.
     */
    function getTotalDistributionAmount() external view returns (uint256) {
        return totalDistributionAmount;
    }

    /**
     * @notice Returns total released amount for all vesting schedules.
     * @return Total released amount for all vesting schedules.
     */
    function getTotalReleasedAmount() external view returns (uint256) {
        return totalReleasedAmount;
    }

    /**
     * @notice Returns a vesting schedule by it's ID. If no vesting schedule
     *         exist with provided ID, returns an empty vesting schedule.
     * @param _vsid An ID of a vesting schedule.
     * @return A vesting schedule structure.
     */
    function getVestingSchedule(
        uint256 _vsid
    ) external view returns (VestingSchedule memory) {
        return vestingSchedules[_vsid];
    }

    /**
     * @notice Returns a list of vesting schedules paginated by their IDs.
     * @param _fromVsid An ID of a vesting schedule to start.
     * @param _toVsid An ID of a vesting schedule to finish.
     * @return A list with the found vesting schedule structures.
     */
    function getPaginatedVestingSchedules(
        uint256 _fromVsid,
        uint256 _toVsid
    ) external view returns (VestingSchedule[] memory) {
        require(_fromVsid <= _toVsid, "Vesting: invalid range");

        uint256 _dataSize = vestingSchedulesCount;

        require(_fromVsid < _dataSize, "Vesting: fromVsid out of bounds");
        require(_toVsid < _dataSize, "Vesting: toVsid out of bounds");

        VestingSchedule[] memory _schedules = new VestingSchedule[](
            _toVsid - _fromVsid + 1
        );

        for (uint256 i = _fromVsid; i <= _toVsid; i++) {
            _schedules[i - _fromVsid] = vestingSchedules[i];
        }

        return _schedules;
    }

    /**
     * @notice Returns releasable amount for a vesting schedule by provided ID.
     *         If no vesting schedule exist with provided ID, returns zero.
     * @param _vsid An ID of a vesting schedule.
     * @return A releasable amount.
     */
    function getReleasableAmount(
        uint256 _vsid
    ) external view returns (uint256) {
        return _computeReleasableAmount(vestingSchedules[_vsid]);
    }

    /**
     * @notice Creates a new vesting schedules in a batch by an owner.
     * @param _accounts An array of addresses of users for whom new vesting
     *                  schedules should be created.
     * @param _amounts An array of vesting schedules distribution amounts.
     * @param _lockPeriods An array of lock period durations (in seconds) that
     *                     should have place before distribution will start.
     * @param _releasePeriods An array of periods (in seconds) during wich ASX
     *                        tokens will be distributed after the lock period.
     */
    function createVestingSchedule(
        address[] memory _accounts,
        uint256[] memory _amounts,
        uint32[] memory _lockPeriods,
        uint32[] memory _releasePeriods
    ) external onlyOwner {
        require(
            _accounts.length > 0,
            "Vesting: accounts array length must be greater than 0"
        );
        require(
            _accounts.length == _amounts.length &&
                _accounts.length == _lockPeriods.length &&
                _accounts.length == _releasePeriods.length,
            "Vesting: lengths mismatch"
        );

        for (uint256 _i = 0; _i < _accounts.length; ++_i) {
            _createVestingSchedule(
                _accounts[_i],
                _amounts[_i],
                _lockPeriods[_i],
                _releasePeriods[_i]
            );
        }
    }

    /**
     * @notice Releases ASX tokens for a specified vesting schedule IDs in a
     *         batch.
     * @param _vsids An array of vesting schedule IDs.
     * @param _recipients An array of recipients of unlocked ASX tokens.
     */
    function release(
        uint256[] memory _vsids,
        address[] memory _recipients
    ) external {
        require(
            _vsids.length > 0,
            "Vesting: vesting schedules IDs array length must be greater than 0"
        );
        require(
            _vsids.length == _recipients.length,
            "Vesting: lengths mismatch"
        );

        for (uint256 _i = 0; _i < _vsids.length; ++_i) {
            _release(_vsids[_i], _recipients[_i]);
        }
    }

    /**
     * @notice Withdraws unused ASX tokens or othe tokens (including ETH) by an
     *         owner.
     * @param _token A token to withdraw. If equal to zero address - withdraws
     *               ETH.
     * @param _amount An amount of tokens for withdraw.
     * @param _recipient A recipient of withdrawn tokens.
     */
    function withdraw(
        address _token,
        uint256 _amount,
        address _recipient
    ) external onlyOwner {
        require(_recipient != address(0), "Vesting: invalid recipient address");

        if (_token == address(0)) {
            payable(_recipient).transfer(_amount);
        } else if (_token != address(token)) {
            IERC20(_token).safeTransfer(_recipient, _amount);
        } else {
            require(
                getWithdrawableASXAmount() >= _amount,
                "Vesting: not enough unused ASX tokens"
            );

            token.safeTransfer(_recipient, _amount);
        }

        emit Withdrawn(_token, _recipient, _amount);
    }

    /**
     * @notice Returns an amount available for withdrawal (unused ASX tokens
     *         amount) by an owner.
     * @return A withdrawable amount.
     */
    function getWithdrawableASXAmount() public view returns (uint256) {
        return token.balanceOf(address(this)) - totalDistributionAmount;
    }

    /**
     * @notice Creates a new vesting schedule.
     * @param _account An address of a user for whom a new vesting schedule
     *                 should be created.
     * @param _amount Vesting schedule distribution amount.
     * @param _lockPeriod A lock period duration (in seconds) that should have
     *                    place before distribution will start.
     * @param _releasePeriod A period (in seconds) during wich ASX tokens will
     *                       be distributed after the lock period.
     */
    function _createVestingSchedule(
        address _account,
        uint256 _amount,
        uint32 _lockPeriod,
        uint32 _releasePeriod
    ) private {
        require(
            getWithdrawableASXAmount() >= _amount,
            "Vesting: not enough unused ASX tokens"
        );
        require(_account != address(0), "Vesting: invalid account address");
        require(
            _amount >= 1 ether,
            "Vesting: amount must be greater than or equal to 1"
        );
        require(_lockPeriod > 0, "Vesting: lock period must be greater than 0");
        require(
            _releasePeriod > 0,
            "Vesting: release period must be greater than 0"
        );

        uint256 _vsid = vestingSchedulesCount;

        vestingSchedules[_vsid] = VestingSchedule(
            _amount,
            0,
            _account,
            uint32(block.timestamp),
            _lockPeriod,
            _releasePeriod
        );

        ++vestingSchedulesCount;

        totalDistributionAmount += _amount;

        emit VestingScheduleCreated(vestingSchedules[_vsid]);
    }

    /**
     * @notice Releases ASX tokens for a specified vesting schedule ID.
     * @param _vsid A vesting schedule ID.
     * @param _recipient A recipient of unlocked ASX tokens.
     */
    function _release(uint256 _vsid, address _recipient) private {
        require(_recipient != address(0), "Vesting: invalid recipient address");

        VestingSchedule memory _vestingSchedule = vestingSchedules[_vsid];

        require(
            _vestingSchedule.startTimestamp != 0,
            "Vesting: vesting schedule does not exist"
        );
        require(
            _vestingSchedule.owner == msg.sender,
            "Vesting: caller is not an owner of a vesting schedule"
        );
        require(
            vestingSchedules[_vsid].released != vestingSchedules[_vsid].amount,
            "Vesting: vesting schedule is ended"
        );

        uint256 _amount = _computeReleasableAmount(_vestingSchedule);

        require(_amount > 0, "Vesting: nothing to release");

        vestingSchedules[_vsid].released += _amount;

        totalReleasedAmount += _amount;

        totalDistributionAmount -= _amount;

        token.safeTransfer(_recipient, _amount);

        emit Released(_vsid, _recipient, _amount);
    }

    /**
     * @notice A method for computing a releasable amount for a vesting
     *         schedule.
     * @param _vestingSchedule A vesting schedule for which to compute a
     *                         releasable amount.
     */
    function _computeReleasableAmount(
        VestingSchedule memory _vestingSchedule
    ) private view returns (uint256) {
        uint32 _currentTime = uint32(block.timestamp);
        uint32 _lockEndTime = _vestingSchedule.startTimestamp +
            _vestingSchedule.lockPeriod;

        if (
            (_currentTime < _lockEndTime) ||
            _vestingSchedule.released == _vestingSchedule.amount
        ) {
            return 0;
        } else {
            uint32 _secondsWithdraw = _currentTime - _lockEndTime;

            if (_secondsWithdraw >= _vestingSchedule.releasePeriod) {
                return _vestingSchedule.amount - _vestingSchedule.released;
            } else {
                return
                    _secondsWithdraw *
                    (_vestingSchedule.amount / _vestingSchedule.releasePeriod) -
                    _vestingSchedule.released;
            }
        }
    }
}